

# Vlastní čísla a vlastní vektory

Připomeňme nejprve definici: Řekneme, že  $\lambda$  je vlastní číslo matice  $\mathbf{A} \in \mathbb{R}^{n \times n}$  a nenulový vektor  $\mathbf{x} \in \mathbb{R}^n$  je odpovídající vlastní vektor, jestliže platí

$$\mathbf{A}\mathbf{x} = \lambda\mathbf{x}.$$

Vlastní čísla reálné matice mohou být reálná nebo komplexní. Symetrická matice má pouze reálná vlastní čísla. Jestliže  $\mathbf{x}$  je vlastní vektor matice  $\mathbf{A}$ , pak i každý jeho nenulový reálný násobek je vlastní vektor matice  $\mathbf{A}$ . Obdobná definice platí i pro matice s komplexními prvky.

Protože

$$\mathbf{A}\mathbf{x} = \lambda\mathbf{x} \iff (\mathbf{A} - \lambda\mathbf{E})\mathbf{x} = \mathbf{0},$$

kde  $\mathbf{E} = \text{diag}(1, \dots, 1)$  je jednotková diagonální matice, a tato homogenní soustava lineárních algebraických rovnic má nenulové řešení právě když matice této soustavy je singulární, počítáme vlastní čísla jako kořeny charakteristické rovnice

$$\det(\mathbf{A} - \lambda\mathbf{E}) = 0.$$

V balíku linalg máme k dispozici následující příkazy:

- `charmat(A, lambda)`;  $\mathbf{A}$  je čtvercová matice,  $\lambda$  je jméno nebo algebraický výraz; funkce **charmat** konstruuje charakteristickou matici  $\mathbf{B} = \lambda\mathbf{E} - \mathbf{A}$ , kde  $\mathbf{E}$  je jednotková matice. Je-li  $\lambda$  jméno, pak  $\det(\mathbf{B})$  je charakteristický polynom matice  $\mathbf{A}$ .
- `charpoly(A, lambda)`;  $\mathbf{A}$  je čtvercová matice řádu  $n$ ,  $\lambda$  je jméno nebo algebraický výraz; funkce **charpoly** počítá charakteristický polynom matice  $\mathbf{A}$ , tj.  $\det(\lambda\mathbf{E} - \mathbf{A})$ , kde  $\mathbf{E}$  je jednotková matice.
- `eigenvalues(A)`; `eigenvalues(A, C)`, kde  $\mathbf{A}$ ,  $\mathbf{C}$  jsou čtvercové matice stejného řádu; příkaz **eigenvalues(A)** vrací posloupnost vlastních čísel matice  $\mathbf{A}$ . Obsahuje-li  $\mathbf{A}$  reálná čísla, je výpočet prováděn numericky s přesností specifikovanou příkazem **Digits**. V symbolickém případě se vlastní čísla počítají jako kořeny charakteristického polynomu. Příkaz **eigenvalues(A, C)** řeší zobecněný problém vlastních čísel, tj. hledá kořeny polynomu  $\det(\lambda\mathbf{C} - \mathbf{A})$ .
- `eigenvectors(A)`, kde  $\mathbf{A}$  je čtvercová matice; procedura **eigenvectors** počítá vlastní čísla a vlastní vektory matice  $\mathbf{A}$ , tj. pro každé vlastní číslo  $\lambda$  řeší homogenní soustavu lineárních algebraických rovnic  $(\mathbf{E}\lambda - \mathbf{A})\mathbf{x} = \mathbf{0}$  pro neznámý vektor  $\mathbf{x}$ . Výsledkem je posloupnost seznamů ve tvaru  $[e_i, \mu, \mathbf{v}_{1,i}, \dots, \mathbf{v}_{n_i,i}]$ , kde  $e_i$  je vlastní číslo,  $\mu$  je jeho algebraická násobnost a  $\mathbf{v}_{1,i}, \dots, \mathbf{v}_{n_i,i}$  jsou vektory, které tvoří bázi vlastního podprostoru odpovídajícího vlastnímu číslu  $e_i$ ,  $1 \leq n_i \leq \mu$ ,  $n_i$  je dimenze vlastního podprostoru. Pro číselné matice se k výpočtu používají standardní numerické metody a výpočet se provádí s přesností specifikovanou příkazem **Digits**. V symbolickém případě se provádí explicitní (přesný) výpočet příslušných vlastních podprostorů.

```
> restart;
> with(linalg): with(plots):
```

Warning, the protected names norm and trace have been redefined and unprotected  
Warning, the name changecoords has been redefined

```
> A := array([[2, 3, -1], [3, -6, 2], [-1, 2, 5]]);
```

$$A := \begin{bmatrix} 2 & 3 & -1 \\ 3 & -6 & 2 \\ -1 & 2 & 5 \end{bmatrix}$$

```
> E := evalm(array(identity,1..3,1..3));
```

$$E := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> B := evalm(A - λ * E);
```

$$B := \begin{bmatrix} 2 - \lambda & 3 & -1 \\ 3 & -6 - \lambda & 2 \\ -1 & 2 & 5 - \lambda \end{bmatrix}$$

```
> charp := det(B);
```

$$\text{charp} := -119 + 46\lambda + \lambda^2 - \lambda^3$$

```
> C:=charmat(A,μ);
```

$$C := \begin{bmatrix} \mu - 2 & -3 & 1 \\ -3 & \mu + 6 & -2 \\ 1 & -2 & \mu - 5 \end{bmatrix}$$

Matici **C** jsme vytvořili příkazem **charmat**. Matice **B** a **C** mají opačná znaménka. Mají stejný charakteristický polynom?

```
> charp1 := det(C);
```

$$\text{charp1} := \mu^3 - \mu^2 - 46\mu + 119$$

Mají polynomy *charp* a *charp1* stejné kořeny? Tedy vedou oba postupy k získání stejných vlastních čísel matice **A**?

Vyzkoušejme ještě příkaz **charpoly**, který počítá charakteristický polynom dané matice **A** jako determinant matice  $\lambda\mathbf{E} - \mathbf{A}$  :

```
> charp2 := charpoly(A, π);
```


$$\text{charp2} := \pi^3 - \pi^2 - 46\pi + 119$$

Na obrázcích 1 a 2 jsou znázorněny polynomy *charp* a *charp2*.


```
> chareq:=det(B) = 0;
```

$$\text{chareq} := -119 + 46\lambda + \lambda^2 - \lambda^3 = 0$$

```
> solve(chareq, {λ});
```



Obrázek 1: Polynom *charp*



Obrázek 2: Polynom *charp2*

$$\left\{ \lambda = \frac{-11188 + 12 I \sqrt{324363}^{(1/3)}}{6} + \frac{278}{3 - 11188 + 12 I \sqrt{324363}^{(1/3)}} + \frac{1}{3} \right\},$$

$$\left\{ \lambda = -\frac{-11188 + 12 I \sqrt{324363}^{(1/3)}}{12} - \frac{139}{3 - 11188 + 12 I \sqrt{324363}^{(1/3)}} + \frac{1}{3} + \right.$$

$$\left. + \frac{1}{2} I \sqrt{3} \left( \frac{-11188 + 12 I \sqrt{324363}^{(1/3)}}{6} - \frac{278}{3 - 11188 + 12 I \sqrt{324363}^{(1/3)}} \right) \right\},$$

$$\left\{ \lambda = -\frac{-11188 + 12 I \sqrt{324363}^{(1/3)}}{12} - \frac{139}{3 - 11188 + 12 I \sqrt{324363}^{(1/3)}} + \frac{1}{3} - \right.$$

$$\left. - \frac{1}{2} I \sqrt{3} \left( \frac{-11188 + 12 I \sqrt{324363}^{(1/3)}}{6} - \frac{278}{3 - 11188 + 12 I \sqrt{324363}^{(1/3)}} \right) \right\}$$

```

> evalf(%);
{λ = 5.435111408 - 0.1 10-8 I}, {λ = -7.395608304 + 0.1339745960 10-9 I},
{λ = 2.960496896 + 0.1866025404 10-8 I}
> λ1 := 5.435111408 - .1e - 8 * I;
      λ1 := 5.435111408 - 0.1 10-8 I
> λ2 := -7.395608304 + .1339745960e - 9 * I;
      λ2 := -7.395608304 + 0.1339745960 10-9 I
> λ3 := 2.960496896 + .1866025404e - 8 * I;
      λ3 := 2.960496896 + 0.1866025404 10-8 I
> evalf(eigenvalues(A));
5.435111408 - 0.1 10-8 I, -7.395608304 + 0.1339745960 10-9 I,
2.960496896 + 0.1866025404 10-8 I

```

Oba postupy nám tedy našly stejná vlastní čísla. Tato vlastní čísla jsou "téměř" reálná. Z obrázku charakteristického polynomu (Obrázek 1, Obrázek 2) vidíme, že kdybychom počítali přesně, měli bychom získat tři reálné kořeny, a tedy tři reálná (různá) vlastní

čísla. Nepřesnost výsledku způsobuje jednak použití numerické metody při výpočtu, jednak zaokrouhlovací chyby.

Zkusme ještě symbolický výpočet například pro diagonální matici:

```
> DM:=array([[a,0,0],[0,b,0],[0,0,c]]);
```

$$DM := \begin{bmatrix} a & 0 & 0 \\ 0 & b & 0 \\ 0 & 0 & c \end{bmatrix}$$

```
> eigenvalues(DM);
```

$a, b, c$

Vidíme, že v případě diagonální matice jsou všechna vlastní čísla diagonální prvky dané matice.

```
> eigenvectors(DM);
```

$[a, 1, \{[1, 0, 0]\}], [b, 1, \{[0, 1, 0]\}], [c, 1, \{[0, 0, 1]\}]$

Výsledek, říká, že  $a$  je jednoduché vlastní číslo, kterému přísluší vlastní vektor  $[1, 0, 0]$ . Obdobně pro vlastní čísla  $b$  a  $c$ .

```
> DM1:=array([[1,0,0],[0,1,0],[0,0,1]]);
```

$$DM1 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> eigenvectors(DM1);
```

$[1, 3, \{[1, 0, 0], [0, 0, 1], [0, 1, 0]\}]$

Tedy matice **DM1** má jedno vlastní číslo rovno 1 s algebraickou násobností 3 a odpovídající vlastní vektory jsou vektory  $(1, 0, 0)^T$ ,  $(0, 1, 0)^T$ ,  $(0, 0, 1)^T$ .

Jaké jsou vlastní vektory matice **A**?

```
> evalf(eigenvectors(A));
```

$[5.435111408 - 0.1 10^{-8} I, 1.,$   
 $\{[-0.179489876 + 0.5105777948 10^{-8} I, 0.127810768 + 0.1252888975 10^{-8} I, 1.]\}$   
 $, [-7.395608304 + 0.1339745960 10^{-9} I, 1.,$   
 $\{[1.79828530 + 0.1847745082 10^{-9} I, -5.298661502 + 0.3593745522 10^{-9} I, 1.]\},$   
 $[2.960496896 + 0.1866025404 10^{-8} I, 1.,$   
 $\{[7.524061680 - 0.4775368097 10^{-8} I, 2.742279288 - 0.1254671347 10^{-8} I, 1.]\}]$

Vlastní vektory matice **A** mají opět "zanedbatelnou" imaginární část:

```
> v1:=vector([1., -.712077918+.5394248717e-9*I,  
> -5.571345162+.2918274615e-8*I]);
```

```
v1 := [1., -0.712077918 + 0.5394248717 10-9 I, -5.571345162 + 0.2918274615 10-8 I]
```

```
> v2:=vector([1., -2.946507676+.1156248156e-9*I,  
> .556085283+.628998508e-10*I]);
```


```
v2 := [1., -2.946507676 + 0.1156248156 10-9 I, 0.556085283 + 0.628998508 10-10 I]
```

```
> v3:=vector([1., .3644679428-.5489823192e-9*I,  
> .132906932-.3662972362e-8*I]);
```


$v_3 := [1., 0.3644679428 - 0.5489823192 \cdot 10^{-9} I, 0.132906932 - 0.3662972362 \cdot 10^{-8} I]$

Znázorníme si ještě některé naše výsledky graficky:


```
> matrixplot(A,axes=boxed);
```



```
> matrixplot(A,heights=histogram,axes=boxed);
```



```
> a1 :=  
> arrow(<Re(v1[1]),Re(v1[2]),Re(v1[3])>, shape=arrow, width=[0.02,  
> relative], head_length=[0.1, relative], color=black):a2 :=  
> arrow(<Re(v2[1]),Re(v2[2]),Re(v2[3])>,shape=arrow, width=[0.02,  
> relative], head_length=[0.2, relative], color=blue): a3 :=  
> arrow(<Re(v3[1]),Re(v3[2]),Re(v3[3])>,shape=arrow, width=[0.1,  
> relative], head_length=[0.3, relative],  
> color=cervena):display(a1,a2,a3, scaling=constrained,  
> axes=FRAMED);
```




Je-li  $\mathbf{v}_1$  vlastní vektor příslušný vlastnímu číslu  $\lambda_1$  matice  $\mathbf{A}$ , pak vektor  $\mathbf{A}\mathbf{v}_1$  je jeho  $\lambda_1$ –násobkem:

```
> w1:=evalm(A&*v1);
```

```
w1 := [5.435111408 - 0.1300000000 10-8 I, -3.870222812 + 0.2600000000 10-8 I,  
-30.28088165 + 0.1567022282 10-7 I]
```

```
> p1 := arrow(<0,0,0>,<Re(v1[1]),Re(v1[2]),Re(v1[3])>), difference,  
> color=black):p2 :=  
> arrow(<0,0,0>,<Re(w1[1]),Re(w1[2]),Re(w1[3])>),shape=arrow,  
> width=[0.02, relative], head_length=[0.2, relative], color=blue):  
> display(p1,p2,axes=FRAMED);
```



### Příklady k procvičení

1. Najděte vlastní čísla a vlastní vektory matice (graficky znázorněte)

$$\mathbf{A} = \begin{pmatrix} 1 & -6 & 3 \\ 1 & 1 & 0 \\ 2 & 0 & 1 \end{pmatrix}.$$

2. Najděte vlastní čísla a vlastní vektory matice (graficky znázorněte)

$$\mathbf{A} = \begin{pmatrix} 1 & 1 & 2 \\ -6 & 1 & 0 \\ 3 & 0 & 1 \end{pmatrix}.$$

3. Najděte vlastní čísla a vlastní vektory matice  $\mathbf{A}$  a matice  $\mathbf{A}^2$  (graficky znázorněte)

$$\mathbf{A} = \begin{pmatrix} 1 & 1 \\ 6 & 0 \end{pmatrix}.$$

4. Najděte vlastní čísla a vlastní vektory matice

$$\mathbf{A} = \begin{pmatrix} 2 & -3 & 1 & -1 \\ 3 & 1 & -2 & 0 \\ -5 & 2 & 0 & 1 \\ 0 & 2 & -1 & 1 \end{pmatrix}.$$

5. Najděte vlastní čísla a vlastní vektory matice

$$\mathbf{A} = \begin{pmatrix} 2 & 1 & 0 & 0 \\ 1 & 0 & -1 & 2 \\ 0 & 0 & 1 & -1 \\ 0 & 1 & 0 & -1 \end{pmatrix}.$$

6. Najděte vlastní čísla a vlastní vektory matice

$$\mathbf{A} = \begin{pmatrix} 2 & 0 & 1 & 0 \\ 0 & -1 & 1 & 0 \\ 1 & 2 & 0 & -1 \\ 0 & -1 & 0 & 1 \end{pmatrix}.$$

7. Najděte vlastní čísla a vlastní vektory matice (graficky znázorněte)

$$\mathbf{A} = \begin{pmatrix} 2 & -1 & 0 \\ 5 & 3 & -2 \\ 4 & -1 & 1 \end{pmatrix}.$$